```
> #Symmetries of the Trefoil knot

> #The trigonometric parametrization is [sin(\theta)+2sin(2\theta),
  cos(\theta)-2cos(2\theta),-sin(3\theta)]
> #By using z=e^{I\theta}=cos(\theta)+I*sin(\theta), with z\in S^1
  (unit circle) we get an alternative, rational, representation [X
  (z),Y(z),Z(z)], with z in the unit circle
> restart:tt:=time():

> X:=simplify((z^2-1)/(2*I*z)+2*(z^4-1)/(2*I*z^2));
X := -((1/2)*I)*(2*z^4+z^3-z-2)/z^2
> Y:=simplify((z^2+1)/(2*z)-2*(z^4+1)/(2*z^2));
Y := -(1/2)*(2*z^4-z^3-z+2)/z^2
> Z:=simplify(-(z^6-1)/(2*I*z^3));
Z := ((1/2)*I)*(z^6-1)/z^3

> #The symmetries satisfy A^[X(z),Y(z),Z(z)]^T+b=[X(\varphi(z)),Y
  (\varphi(z)),Z(\varphi(z))]^T, with A a 3x3 regular, orthogonal
  matrix, b a real vector in 3-space, and \varphi(z)=-z or \varphi
  (z)=k/z with k a complex number of modulus equal to 1.
> #We first try with \varphi(z)=-z

> C:=[X,Y,Z]:
> Eq[1]:=numer((m11*C[1]+m12*C[2]+m13*C[3]+b1)-subs(z=-z,C[1])):
> Eq[2]:=numer((m21*C[1]+m22*C[2]+m23*C[3]+b2)-subs(z=-z,C[2])):
> Eq[3]:=numer((m31*C[1]+m32*C[2]+m33*C[3]+b3)-subs(z=-z,C[3])):
> n:=max(degree(Eq[1],z),degree(Eq[2],z),degree(Eq[2],z));
n := 6
> lis:=[]:
> for i from 1 to 3 do
    for j from 0 to n do
      lis:=[op(lis),coeff(Eq[i],z,j)]:
    od:
  od:
> with(SolveTools);
[AbstractRootOfSolution, Basis, CancelInverses, Combine,
Complexity, Engine, GreaterComplexity, Identity, Inequality,
Linear, Parametric, Polynomial, PolynomialSystem,
RationalCoefficients, SemiAlgebraic, SortByComplexity]
> PolynomialSystem(lis,[m11,m12,m13,m21,m22,m23,m31,m32,m33,b1,b2,
  b3],engine=groebner);
> #No solution here
> #Now we try with \varphi(z)=k/z, with k in S^1

> Eq[1]:=numer((m11*C[1]+m12*C[2]+m13*C[3]+b1)-subs(z=k/z,C[1])):
> Eq[2]:=numer((m21*C[1]+m22*C[2]+m23*C[3]+b2)-subs(z=k/z,C[2])):
> Eq[3]:=numer((m31*C[1]+m32*C[2]+m33*C[3]+b3)-subs(z=k/z,C[3])):
> n:=max(degree(Eq[1],z),degree(Eq[2],z),degree(Eq[2],z));
n := 7
> lis:=[]:
```

```
> for i from 1 to 3 do
    for j from 0 to n do
      lis:=[op(lis),coeff(Eq[i],z,j)]:
    od:
  od:
> with(SolveTools);
```

```
[AbstractRootOfSolution, Basis, CancelInverses, Combine,
Complexity, Engine, GreaterComplexity, Identity, Inequality,
Linear, Parametric, Polynomial, PolynomialSystem,
RationalCoefficients, SemiAlgebraic, SortByComplexity]
```

```
> PolynomialSystem(lis,[m11,m12,m13,m21,m22,m23,m31,m32,m33,b1,b2,
  b3,k],engine=groebner);
```

```
{b1 = 0, b2 = 0, b3 = 0, k = 1, m11 = -1, m12 = 0, m13 = 0, m21
= 0, m22 = 1, m23 = 0, m31 = 0, m32 = 0, m33 = -1}, {b1 = 0, b2
= 0, b3 = 0, k = RootOf(_Z^2+_Z+1), m11 = 1/2, m12 = -I*RootOf
(_Z^2+_Z+1)-(1/2)*I, m13 = 0, m21 = -I*RootOf(_Z^2+_Z+1)-(1/2)*
I, m22 = -1/2, m23 = 0, m31 = 0, m32 = 0, m33 = -1}
```

```
> #Here we have three solutions; one can check that they correspond
  to rotational symmetries about the z-axis by angles ±2π/3, as
  well as a half-turn about the y-axis
> time()-tt;
```

```
0.63e-1
```

```
> with(plots):p1:=spacecurve([sin(theta)+2*sin(2*theta),cos(theta)
  -2*cos(2*theta),-sin(3*theta)],theta=-6..6,color=blue,thickness=
  3,numpoints=50000):
```

```
> display(p1);#Plot of the curve
```



−3
−2
−1
0
1
2
−1
−0.5
0
0.5
1
−2
−1
0
1
2